Obtaining Laurent Series & residues using *Mathematica*

Laurent Series example discussed in Boas and in class

In[343]:=
```
Clear[ff]
```

In[344]:=
```
ff[z_] = 12 / (z (2 - z) (1 + z))
```

Out[344]=
$$\frac{12}{(2-z)\ z\ (1+z)}$$

Inner region R1

Mathematica command Series[] automatically gives Laurent series.
{z,0,3} means: expand in z, about z=0, giving up to z^3 term.

In[345]:=
```
Series[ff[z], {z, 0, 3}]
```

Out[345]=
$$\frac{6}{z} - 3 + \frac{9\,z}{2} - \frac{15\,z^2}{4} + \frac{33\,z^3}{8} + O[z]^4$$

Outermost region R3

Here we expand about z=Infinity, and *Mathematica* automatically does the series in powers of 1/z

In[347]:=
```
Series[ff[z], {z, Infinity, 6}]
```

Out[347]=
$$-\frac{12}{z^3} - \frac{12}{z^4} - \frac{36}{z^5} - \frac{60}{z^6} + O\left[\frac{1}{z}\right]^7$$

Middle region R2

Partial fraction function:

In[372]:=
```
Clear[ff1, ff2]
```

In[373]:=
```
ff1[z_] = 4 / (z (1 + z)); ff2[z_] = 4 / (z (2 - z));
```

Check that sum agrees with original function

In[374]:=
```
FullSimplify[ff1[z] + ff2[z] - ff[z]]
```

Out[374]= 0

In[379]:=
```
ff1expand = Normal[Series[ff1[z], {z, Infinity, 4}]]
```

Out[379]=
$$\frac{4}{z^4} - \frac{4}{z^3} + \frac{4}{z^2}$$

In[380]:=
```
ff2expand = Normal[Series[ff2[z], {z, 0, 3}]]
```

Out[380]=
$$1 + \frac{2}{z} + \frac{z}{2} + \frac{z^2}{4} + \frac{z^3}{8}$$

In[381]:=
```
result = ff1expand + ff2expand
```

Out[381]= $1+\frac{4}{z^4}-\frac{4}{z^3}+\frac{4}{z^2}+\frac{2}{z}+\frac{z}{2}+\frac{z^2}{4}+\frac{z^3}{8}$

Residues

This is a built-in *Mathematica* command.
Here is the example of the function discussed above.

In[382]:=
```
Residue[ff[z], {z, 0}]
```

Out[382]= 6

In[384]:=
```
Residue[ff[z], {z, -1}]
```

Out[384]= -4

In[385]:=
```
Residue[ff[z], {z, 2}]
```

Out[385]= -2